Maple V (and Maple 6)

This report is taken from Appendix D of the second edition of the textbook – slightly revised for the web. It was written and fully tested at the time based on *Maple V*. Now, our notes from when we were using *Maple 6* for our classes indicate that *Maple V* and *Maple 6* are almost identical with regards to the operations discussed here. The one exception of which we are aware is noted below. However, we no longer had *Maple 6* available to thoroughly test at the time this revised report was written. Thus, it is possible that *Maple 6* might produce output in a slightly different format than shown here, but any differences should be insignificant.

You are prompted for each line of input in *Maple V* with the ">" symbol, and each input line must be followed by a semicolon. Users of *Maple V* who want to perform linear algebra computations should begin by loading a utility named "linalg." This is done simply by typing

```
with(linalg);
```

and we assume this has been done for all that follows.

Input of Vectors and Matrices; Fundamental Operations

To enter a vector, use the `vector` command. Type the word `vector`, and then, within parentheses, type a pair of brackets containing the vector entries (separated by commas). To enter a matrix, use the `matrix` command. First indicate the number of rows and the number of columns, followed by a pair of brackets "`[ ]`" containing the entries (row by row). It is often useful to assign names to vectors and matrices as they are entered. Use a colon followed by an equal sign ("`:=`") to give a symbolic name to a vector or matrix.

Figures D.1 and D.2 illustrate the input of vectors/matrices as well as the fundamental operations of dot product, scalar multiplication, matrix multiplication, transpose, and inverse. The `evalm` function evaluates the result of vector/matrix calculations. The `dotprod` function calculates the dot product of two vectors. Addition and subtraction are performed using "+" and "−", respectively. Scalar multiplication can never be implied simply by putting a scalar next to a vector; instead, you must use "`*`". Matrix multiplication is indicated by "`&*`." The "`^`" symbol is used to find integer powers of a square matrix. In particular, "`^`(−1)" is used to calculate the inverse of a square matrix. The `transpose` function calculates the transpose of a matrix.

The number of significant digits displayed by *Maple V* is controlled by the value of the variable `Digits`. *Maple V* attempts to express all calculations in fractional form if possible. To force an expression into decimal form, simply include a decimal number somewhere in the expression, as in the second expression for matrix `M5` in Figure D.2.

Input	Output
`>with(linalg);`	(opens linear algebra utility)
`>v1:=vector([5,7,-4]);`	$v1 := [5, 7, -4]$
`>v2:=vector([x,y,z]);`	$v2 := [x, y, z]$
`>v3:= evalm(2*v1+3*v2);`	$v3 := [10 + 3x, 14 + 3y, -8 + 3z]$
`>dotprod(v1,v2);`	$5x + 7y - 4z$

Figure D.1: *Maple V* session: vectors; fundamental vector operations

Input	Output
`>M1:=matrix(3,4,[4,-1,6,-2, -3,2,-3,2,-6,8,1,3]);`	$M1 := \begin{bmatrix} 4 & -1 & 6 & -2 \\ -3 & 2 & -3 & 2 \\ -6 & 8 & 1 & 3 \end{bmatrix}$
`>M2:=matrix(4,3,[2,-3,0,6,8,-1, 3,1,-2,2,-4,-2]);`	$M2 := \begin{bmatrix} 2 & -3 & 0 \\ 6 & 8 & -1 \\ 3 & 1 & -2 \\ 2 & -4 & -2 \end{bmatrix}$
`>M3:=evalm(M1-2*transpose(M2));`	$M3 := \begin{bmatrix} 0 & -13 & 0 & -6 \\ 3 & -14 & -5 & 10 \\ -6 & 10 & 5 & 7 \end{bmatrix}$
`>M4:=evalm(M1&*M2);`	$M4 := \begin{bmatrix} 16 & -6 & -7 \\ 1 & 14 & 0 \\ 45 & 71 & -16 \end{bmatrix}$
`>M5:=evalm(M4^(-1));`	$M5 := \begin{bmatrix} \frac{-224}{233} & \frac{-593}{233} & \frac{98}{233} \\ \frac{16}{233} & \frac{59}{233} & \frac{-7}{233} \\ \frac{-559}{233} & \frac{-1406}{233} & \frac{230}{233} \end{bmatrix}$
`>Digits:=4;`	*Digits* := 4;
`>M5:=evalm(1.0*M4^(-1));`	$M5 := \begin{bmatrix} -.9614 & -2.545 & .4206 \\ .06867 & .2532 & -.03004 \\ -2.399 & -6.034 & .9871 \end{bmatrix}$

Figure D.2: *Maple V* session: matrices; fundamental matrix operations

Solving a Linear System; Gauss-Jordan Row Reduction Method

You can solve a linear system using the `solve` function, by simply listing a set (in braces such as "`{ }`") containing each equation in the system in turn. When printing out the solution set of such a system, *Maple V* expresses each dependent variable in terms of the independent variables. If a system has no solution, *Maple V* returns no output at all.

You can also solve a linear system using the `rref` function. This function calculates the reduced row echelon form of a (possibly augmented) matrix. The matrix `M6` in Figure D.3 is the augmented matrix for a linear system with an infinite solution set.

The `solve` and `rref` functions are illustrated in Figure D.3, in which the same linear system is solved using both commands. Using either method, you can easily see that the general solution set is $\{(-3c+4,\ 2c+5,\ c,\ -2)\}$.

Input	Output
`>solve({3*x+y+7*z+2*w=13, 2*x-4*y+14*z-w=-10, 5*x+11*y-7*z+8*w=59, 2*x+5*y-4*z-3*w=39});`	$\{x=-3z+4,\ y=2z+5,\ w=-2,\ z=z\}$
`>M6:=matrix(4,5,[3,1,7,2,13, 2,-4,14,-1,-10,5,11,-7,8,59, 2,5,-4,-3,39]);`	$M6 := \begin{bmatrix} 3 & 1 & 7 & 2 & 13 \\ 2 & -4 & 14 & -1 & -10 \\ 5 & 11 & -7 & 8 & 59 \\ 2 & 5 & -4 & -3 & 39 \end{bmatrix}$
`>M7:=rref(M6);`	$M7 := \begin{bmatrix} 1 & 0 & 3 & 0 & 4 \\ 0 & 1 & -2 & 0 & 5 \\ 0 & 0 & 0 & 1 & -2 \\ 0 & 0 & 0 & 0 & 0 \end{bmatrix}$

Figure D.3: *Maple V* session: solution of a linear system; row reduction

Determinants; Eigenvalues/Eigenvectors; Characteristic Polynomial

The `det` function calculates the determinant of a square matrix. The characteristic polynomial for a square matrix `M` can be found by using the function `charpoly(M,`x`)`. Eigenvalues for a square matrix can be found by factoring the characteristic polynomial using the function `factor`. The quotation mark in *Maple V* represents the most recent output, and so the command `factor(")` indicates that the previous result is to be factored. Note that in *Maple 6*, the `%` sign is used to indicate the most recent output instead of the quotation mark. Hence, the command `factor(%)` would be used to factor the previous result in *Maple 6*.

A more direct way to calculate the eigenvalues is to use the `eigenvals` function. The `eigenvects` function is even more powerful: it lists each eigenvalue, its algebraic multiplicity, and a basis for its eigenspace, respectively. Most of these functions are illustrated in Figure D.4, where the matrix `M8` has two eigenvalues, $\lambda_1 = -5$, having algebraic and geometric multiplicity 1, and $\lambda_2 = 3$, having algebraic multiplicity 3 and geometric multiplicity 2.

Input	**Output**
`>M8:=matrix(4,4,[5,2,0,1,-2,1,0,-1,` `4,4,3,2,16,0,-8,-5]);`	$M8 := \begin{bmatrix} 5 & 2 & 0 & 1 \\ -2 & 1 & 0 & -1 \\ 4 & 4 & 3 & 2 \\ 16 & 0 & -8 & -5 \end{bmatrix}$
`>charpoly(M8,x);`	$x^4 - 4x^3 - 18x^2 + 108x - 135$
`>factor(");` (in *Maple V*) `>factor(%);` (in *Maple 6*)	$(x+5)(x-3)^3$
`>eigenvals(M8);`	$-5, 3, 3, 3$
`>eigenvects(M8);`	$[-5, 1, \{-1, 1, -2, 8]\}]$, $[3, 3, \{[1, 0, 4, -2], [0, 1, 2, -2]\}]$

Figure D.4: *Maple V*/*Maple 6* session: characteristic polynomial; eigenvalues; eigenvectors

The characteristic polynomial of a matrix $\mathbf{M}$ can also be computed directly by calculating the determinant of $x\mathbf{I}_n - \mathbf{M}$. The identity matrix is represented by "`&*()`", where *Maple V* determines its correct size from context.

Of course, a basis of eigenvectors for each eigenvalue λ of a square matrix `M` can also be calculated by row reducing the matrix $\lambda\mathbf{I}_n-$ `M`, setting each independent variable in turn equal to 1 with all others equal to 0, and then solving for the dependent variables.

These operations are illustrated in Figure D.5. Linearly independent eigenvectors for the earlier matrix `M8`, for eigenvalue $\lambda_2 = 3$, are found from the reduced row echelon form matrix `M10` for `M9` $= 3\mathbf{I}_4 -$ `M8`. First, by letting the third column variable of `M10` equal 1 and its fourth column variable equal 0, we obtain $[\frac{1}{2}, -\frac{1}{2}, 1, 0]$, and then by letting its third column variable equal 0 and fourth column variable equal 1, we obtain $[\frac{1}{2}, -1, 0, 1]$. (You can easily verify that $\{[\frac{1}{2}, -\frac{1}{2}, 1, 0], [\frac{1}{2}, -1, 0, 1]\}$ spans the same two-dimensional subspace of $\mathbb{R}^4$ as the set $\{[1, 0, 4, -2], [0, 1, 2, -2]\}$ of eigenvectors obtained earlier from the `eigenvects` function.)

Input	Output
`>det(x*&*( )-M8);`	$x^4 - 4x^3 - 18x^2 + 108x - 135$
`>factor(");` (in *Maple V*) `>factor(%);` (in *Maple 6*)	$(x+5)(x-3)^3$
`>M9:=evalm(3*&*()-M8);`	$M9 := \begin{bmatrix} -2 & -2 & 0 & -1 \\ 2 & 2 & 0 & 1 \\ -4 & -4 & 0 & -2 \\ -16 & 0 & 8 & 8 \end{bmatrix}$
`>M10:=rref(M9);`	$M10 := \begin{bmatrix} 1 & 0 & -\frac{1}{2} & -\frac{1}{2} \\ 0 & 1 & \frac{1}{2} & 1 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 \end{bmatrix}$

Figure D.5: *Maple V*/*Maple 6* session: characteristic polynomial via determinant; direct calculation of eigenspace

Gram-Schmidt Process

You can perform the Gram-Schmidt Process in *Maple V* for a given set of vectors simply by using the fundamental operations for vector addition and scalar multiplication. However, there is a built-in function, `GramSchmidt` (no hyphen), that replaces a given set of linearly independent vectors with an orthogonal basis for the span of those vectors. (Note: The orthogonal basis obtained is not necessarily the same basis that you will obtain using the formulas for the Gram-Schmidt Process in Section 6.1 of the textbook.) An orthonormal basis for the span can easily be produced by dividing each orthogonal basis vector by its length. The function `norm(v,2)` calculates the length of the vector `v`. These operations are illustrated in Figure D.6, where we find an orthogonal, and then an orthonormal, basis for the subspace of $\mathbb{R}^4$ spanned by the vectors labeled `v5`, `v6`, and `v7`.

Input	Output
`>v5:=vector([2,1,0,-1]);`	$v5 := [2, 1, 0, -1]$
`>v6:=vector([1,0,2,-1]);`	$v6 := [1, 0, 2, -1]$
`>v7:=vector([0,-2,1,0]);`	$v7 := [0, -2, 1, 0]$
`>GramSchmidt({v5,v6,v7});`	$\left\{[1, 0, 2, -1], \left[-\frac{1}{3}, -2, \frac{1}{3}, \frac{1}{3}\right], \left[\frac{33}{26}, -\frac{5}{13}, -\frac{10}{13}, -\frac{7}{26}\right]\right\}$
`>v8:=evalm(v6/norm(v6,2));`	$v8 := \left[\frac{1}{6}\sqrt{6}, 0, \frac{1}{3}\sqrt{6}, -\frac{1}{6}\sqrt{6}\right]$
`>v9:=vector([-1/3,-2,1/3,1/3]);`	$v9 := \left[-\frac{1}{3}, -2, \frac{1}{3}, \frac{1}{3}\right]$
`>v10:=evalm(v9/norm(v9,2));`	$v10 := \left[-\frac{1}{39}\sqrt{39}, -\frac{2}{13}\sqrt{39}, \frac{1}{39}\sqrt{39}, \frac{1}{39}\sqrt{39}\right]$
`>v11:=vector([33/26,-5/13,-10/13,-7/26]);`	$v11 := \left[\frac{33}{26}, -\frac{5}{13}, -\frac{10}{13}, -\frac{7}{26}\right]$
`>v12:=evalm(v11/norm(v11,2));`	$v12 := \left[\frac{11}{182}\sqrt{182}, -\frac{5}{273}\sqrt{182}, -\frac{10}{273}\sqrt{182}, -\frac{1}{78}\sqrt{182}\right]$

Figure D.6: *Maple V* session: Gram-Schmidt Process; orthogonal and orthonormal bases

You can easily verify that $\left\{[1, 0, 2, -1], \left[-\frac{1}{3}, -2, \frac{1}{3}, \frac{1}{3}\right], \left[\frac{33}{26}, -\frac{5}{13}, -\frac{10}{13}, -\frac{7}{26}\right]\right\}$ (vectors `v6`, `v9`, `v11`) is an orthogonal set of vectors spanning the same subspace of $\mathbb{R}^4$ as $\{[2, 1, 0, -1], [1, 0, 2, -1], [0, -2, 1, 0]\}$. An orthonormal basis for the same subspace is given by the vectors `v8`, `v10`, and `v12`.